Lab 5

In this lab, we will learn different ways to draw parametric curves, polar curves, and conic sections.

- Plot a curve using `ParametricPlot`,
- Plot a polar curve using `PolarPlot`,
- Shade the region enclosed by curves using `ParametricPlot`,
- Draw a conic section using `ParametricPlot`, `PolarPlot`, and `ContourPlot`.

1 Plot a curve using ParametricPlot

The command **ParametricPlot** can go with different syntaxes depending on which functionalities you want to use. Here, we will only look into two main syntaxes/functionalities*. The first syntax is

```
ParametricPlot[{x,y}, {t,a,b}]
```

which draws a *curve* consisting of points (x, y) where x and y are functions of $t \in [a, b]$. The second syntax is

```
ParametricPlot[{x,y}, {t,a,b}, {s,c,d}]
```

which draws a *region* consisting of points (x, y) where x and y are functions of $t \in [a, b]$ and $s \in [c, d]$. Drawing regions will be discussed in Section 3.

(1) To draw the curve parametrized by $x = t \sin t$, $y = \cos t$, try the following

```
ParametricPlot[{t*Sin[t], Cos[t]}, {t, -5, 5}]
```

Change the color of the curve by putting an option `PlotStyle -> Red` inside the command.

(2) Alternatively, you can break this command into:

```
x[t_] := t*Sin[t];
y[t_] := Cos[t];
ParametricPlot[{x[t], y[t]}, {t, -5, 5}]
```

This way is advantageous if x and y are cumbersome expressions (for example, Exercise (5)).

(3) Recall that the parametric equation of a curve informs not only the shape of the curve but also how the curve is drawn. You can create an interactive plot using **Manipulate** to see Mathematica draws this curve. This command works best on the desktop version of Mathematica (compared to the cloud version). To see how Mathematica draws the curve in the previous exercise, try the following:

```
p[s_] :=  ParametricPlot[{t*Sin[t], Cos[t]}, {t, -5, s},
            PlotRange -> {{-5, 2}, {-1, 1}}]
Manipulate[p[s], {s, -4.9, 5}]
```

*If you are interested in learning more functionalities of the command `ParametricPlot`, check out Wolfram's Documentation website: https://reference.wolfram.com/language/ref/ParametricPlot.html

On the desktop version of Mathematica, you will see next to the plot a control bar with a Play button. If you click on it, you will see dynamic plotting as s moves from -4.9 to 5. Some dynamic plotting features are not available on the cloud version of Mathematica.

(4) In the above command, the option `PlotRange` is to fix the plot window, which is $x \in [-5, 2]$ and $y \in [-1, 1]$ in this case. Remove the `PlotRange` option from the command, and run the command again. Allow s to run from -4.9 to 5 by clicking on the Play button or by dragging the cursor with your mouse. Do you see anything different than before?

(5) Graph the curve parametrized by

$$\begin{aligned} x &= (e^{\cos t} - 2\cos(4t) - \sin^5(t/12))\sin t \\ y &= (e^{\cos t} - 2\cos(4t) - \sin^5(t/12))\cos t \end{aligned}$$

where $t \in [0, 20]$. Then use the command `Manipulate` to see how Mathematica plots this curve.

(6) Repeat the above exercise but with

$$\begin{aligned} x &= \sqrt{3}\cos(2t) - \cos(10t)\sin(20t) \\ y &= \sqrt{2}\sin(2t) - \sin(10t)\sin(20t) \end{aligned}$$

for $t \in [0, \pi]$.

(7) A *cycloid* is the curve traced by a point on the rim of a circular wheel rolling along a straight line. Suppose the radius of the wheel is a. Then the parametric equation of the cycloid is

$$x = a(t - \sin t), \quad y = a(1 - \cos t)$$

Graph three cycloids with the wheel radii $a = 1$, $a = 1.5$, $a = 2$ on the same plot. You can plot each cycloid individually with a different color and combine them together using the command `Show`. You might need to specify a plot range in each command so that the combined plot is not cut off.

2 Plot a polar curve using PolarPlot

A polar curve is simply a curve given by an equation $r = r(\theta)$ in the polar coordinates. To make the notation more convenient, let us rename θ by t. The command **PolarPlot** can be used to draw a polar curve. To draw a polar curve $r = r(t)$, the syntax is

```
PolarPlot[r[t], {t,a,b}]
```

(8) For example, to plot the polar curve $r = 1/t$ for $t \in [1, 50]$, try the following:

```
PolarPlot[1/t, {t,1,50}, PlotRange -> Full]
```

Can you guess if, as t increases, the spiral goes inside out or outside in? Explain your answer.

(9) Use the command `Manipulate` to see how Mathematica draws this spiral.

(10) Recall that there is a relation between polar coordinates (r, t) and Cartesian coordinates (x, y):

$$x = r\cos t, \quad y = r\sin t$$

Therefore, the curve in Exercise (8) can be described in Cartesian coordinates as

$$x = \frac{1}{t}\cos t, \quad y = \frac{1}{t}\sin t$$

Use the command `ParametricPlot` to draw the curve.

(11) Plot the polar curve $r(t) = \sin(nt)$ with $n = 1, 2, 3, 4, 5, 6, 7$. For a general n, can you guess how many petals the flower has?

(12) Plot the polar curve $r(t) = 1 + \cos(nt) + \sin^2(nt)$ with $n = 2, 3, 4, 5$. For which value of n do you get a shamrock leaf?

3 Shade the region enclosed by curves

In Lab 1, you learned how to use `Filling` to shade the region between curves. It is quite tricky to use `Filling` to shade the region enclosed by a polar curve or between two polar curves. Here, you will learn a more effective method to do so using `ParametricPlot`. Specifically, you will use the following command

```
ParametricPlot[{x,y}, {t,a,b}, {s,c,d}]
```

to draw a *region* consisting of points (x, y) where x and y are functions of $t \in [a, b]$ and $s \in [c, d]$.

(13) For example, any point lying under the parabola $y = x^2$, $0 \leq x \leq 2$, and above the x-axis has coordinates (x, y) where $0 \leq x \leq 2$, $0 \leq y \leq x^2$. Such a point (x, y) can be represented as

$$x = t, \quad y = st^2$$

where $0 \leq t \leq 2$ and $0 \leq s \leq 1$. Use the following command to sketch the region:

```
ParametricPlot[{t, s t^2}, {t,0,2}, {s,0,1}]
```

(14) More generally, y is in between two numbers N and M, then one can write $y = N + s(M - N)$ for $0 \leq s \leq 1$. (In the previous exercise, $N = 0$ and $M = t^2$ or vice versa.) Use this rule and the command `ParametricPlot` to shade the region between two curves $y = \cos x$ and $y = \sin x$ for $0 \leq x \leq 8\pi$.

(15) Draw the polar curve $r = 1 + \cos t$. This curve is called a *cardioid* (a heart shape).

(16) To shade the region enclosed by this cardioid, note that every point inside the cardioid has polar coordinates (r, t) satisfying $0 \leq t \leq 2\pi$ and r in between 0 and $1 + \cos t$. Thus, $r = s(1 + \cos t)$ with $0 \leq s \leq 1$. Therefore, any point inside the curve has Cartesian coordinates

$$x = s(1 + \cos t)\cos t, \qquad y = s(1 + \cos t)\sin t$$

with $0 \leq t \leq 2\pi$ and $0 \leq s \leq 1$. To sketch this region, try the following command:

```
ParametricPlot[{s(1 + Cos[t])Cos[t], s(1 + Cos[t])Sin[t]},
              {t, 0, 2 Pi}, {s, 0, 1}, PlotPoints -> 50]
```

(17) Alternatively, you can break the command into

```
r[t_] := 1 + Cos[t];
ParametricPlot[{s*r[t]*Cos[t], s*r[t]*Sin[t]},
    {t, 0, 2 Pi}, {s, 0, 1}, PlotPoints -> 50]
```

This way of breaking down the command is helpful when r is a cumbersome expression of t, such as in the following exercise.

(18) On the same plot, draw the cardioid $r = 1 + \cos t$ and shade the section inside the cardioid with $0 \le t \le \pi/2$. Hint: draw the cardioid and shade the section separately, and then combine them using the command `Show`.

(19) Draw the polar curve

$$r = \frac{100}{100 + \left(t - \frac{\pi}{2}\right)^8}\left(2 - \sin(7t) - \frac{\cos(30t)}{2}\right)$$

for $t \in [-\frac{\pi}{2}, \frac{3\pi}{2}]$ and then shade the region enclosed by the curve using `ParametricPlot`.

(20) On the same plot, draw the maple leaf in the previous exercise and shade the section inside the curve with $\pi/6 \le t \le 5\pi/6$.

(21) Recall that the length of the polar curve $r = r(t)$, $a \le t \le b$ is

$$L = \int_a^b \sqrt{r^2 + (r')^2}dt$$

and the area swept by the polar curve from $t = a$ to $t = b$ is

$$A = \int_a^b \frac{1}{2}r^2 dt.$$

Find the total length of the maple leaf in Exercise (19) and the area enclosed by it.

(22) Find the area and perimeter of the shaded region in Exercise (20). The perimeter consists of two straight edges and one curve.

(23) Consider a four-petaled flower $r = \cos(2t)$. Shade one of the petals (of your choice). Then find the length and area of that petal.

4 Draw a conic section

Recall that a conic section can be given by an equation in Cartesian coordinates or or an equation in polar coordinates. For example, the equation

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

describes an ellipse, and the equation

$$\frac{x^2}{a^2} - \frac{y^2}{b^2} = 1$$

describes a hyperbola. In polar coordinates, all conic sections with the exception of the circle can be given by the equation

$$r = \frac{ed}{1 \pm e\cos t} \quad \text{or} \quad r = \frac{ed}{1 \pm e\sin t} \tag{1}$$

(24) The command `ContourPlot` is used to plot an equation. For example, to plot the ellipse $x^2 + y^2/4 = 6$, try the following

```
ContourPlot[x^2 + y^2/4 == 6, {x, -3, 3}, {y, -5, 5},
     AspectRatio -> Automatic, Axes -> True, AxesLabel -> Automatic]
```

(25) Try adding `ContourStyle -> {Thick, Red}` into the above command.

(26) The ellipse above has a standard form

$$\frac{x^2}{6} + \frac{y^2}{24} = 1$$

and thus, the focal length is $c = \sqrt{24-6} = \sqrt{18}$. The coordinates of the foci are $(0, \pm\sqrt{18})$ and the equations of the directrix are $y = \pm d = \pm\frac{24}{\sqrt{18}}$. You can draw the ellipse, the foci, and the two directrix on the same plot as follows:

```
c = Sqrt[18];
d = 24/c;
p1 = ContourPlot[x^2 + y^2/4 == 6, {x, -3, 3}, {y, -5, 5},
     AspectRatio -> Automatic, Axes -> True, AxesLabel -> Automatic,
         ContourStyle -> {Thick, Red}];
p2 = ContourPlot[{y == d, y == -d}, {x, -3, 3}, {y, -6, 6}];
p3 = ListPlot[{{0, -c}, {0, c}}, PlotStyle -> PointSize[Large]];
Show[p1, p2, p3, PlotRange -> All]
```

(27) On the same plot, draw the hyperbola

$$\frac{y^2}{4} - \frac{x^2}{8} = 4$$

together with the foci, the directrix lines, and the asymptotes.

(28) On the same plot, draw the parabola

$$x^2 - 2x = 4y + 7$$

together with the focus and the directrix line.

(29) Use Equation (1) and the command `PolarPlot` to sketch the ellipse with $e = 0.7$ and $d = 1$. Try all 4 different formulas of Equation (1) (corresponding to the choice of plus/minus sign and cosine/sine). What differences do you observe in the 4 pictures?

(30) Use the formula $r = \frac{ed}{1+e\cos t}$ with $t \in [0, 2\pi]$ and the command `PolarPlot` to draw a hyperbola with $e = 3$ and $d = 1$. Does it draw only one branch of the hyperbola or both branches? Can you explain why?

5 To turn in

Submit your implementation of Exercises (1) - (30) as a source file (nb/ipynb) and a pdf file.